
l na dużo więcej niż uda mi się opisać – sam jej opis zajmuje 25 stron 

dokumentacji! Z tego też powodu opisuję jedynie najprostszą apli-
kację tej magistrali. Pełna dokumentacja tej i innych magistral Ava-
lon dostępna jest w dokumencie Avalon Interface Specifications.

W  tabeli  1 umieszczono uproszczony opis tych linii (wraz 
ze standardowymi liczbami bitów (szerokościami), oraz ich kie-
runkiem, w odniesieniu do układu slave), które przyjdzie nam 
wkrótce zdefiniować. Każdy układ na magistrali taktowany jest 

NIOS II na maXimatorze, 
czyli mikroprocesor 
w układzie FPGA (9)
Własne moduły w systemie – wprowadzenie

Nadszedł wreszcie czas na to, aby przygotować się do zastosowania największej zalety FPGA w porównaniu 
z mikrokontrolerami – możliwości implementacji własnych układów logicznych. Żeby móc łączyć takie układy 
z naszym systemem NIOS II, musimy poznać podstawową magistralę w nim stosowaną – Avalon Memory Map-
ped (Avalon-MM). Aby jednak nasza nauka nie była czysto teoretyczną, wykonamy implementację generatora 
PWM, czyli elementu, który świetnie znamy z mikrokontrolerów, ale który nie jest dostępny w formie gotowego 
modułu. Ponadto ta część kursu i kolejne prowadzone będą w środowisku Quartus Lite 18.0 – tak, tak – kolejna 
aktualizacja, tym razem bez większych rewolucji.

Magistrala Avalon-MM służy do transmisji danych w dwóch kie-
runkach, z możliwością adresacji. Zachowuje się więc podobnie 
jak pamięć np. RAM. Stąd też jej nazwa (Memory Mapped). Na ma-
gistrali tej są wyróżnione układy (u nas będzie to  jeden układ) 
nadrzędne (master) oraz podrzędne (slave). Magistrala ta w podsta-
wowej konfiguracji, którą zaraz poznamy, zawiera osobne porty rów-
noległe do transmisji danych z i do układu master, port adresowy 
oraz dodatkowe linie sterujące. Magistrala Avalon-MM pozwala 

Tabela 1. Opis podstawowych linii magistrali Avalon-MM
Sygnał Liczba

bitów
Kierunek Opis

Avalon-MM
address 0-30 In Magistrala adresująca słowa („rejestry”) w tworzonym układzie slave. W naszym systemie słowa 

są 32-bitowe, czyli ustawienie w danym module szerokości tej magistrali na 2 bity spowoduje, że do-
stępne będziemy mieli 4 słowa 32-bitowe.

byteenable 4 In Umożliwia wybór zapisu jedynie określonych bajtów tworzących słowo tak, aby pomimo używania 
32-bitowych słów możliwe było adresowanie bajtów ze strony układu master. Działanie tych linii zo-
stanie omówione szczegółowo nieco później. Linie te mają znaczenie tylko dla zapisu do układu slave.

read 1 In Stan wysoki na tej linii oznacza, że w danym cyklu wykonywany jest odczyt, czyli dane spod określo-
nego adresu powinny zostać wystawione na magistralę readdata.

readdata 32 Out Magistrala odczytu danych
write 1 In Stan wysoki na tej linii oznacza, że w danym cyklu wykonywany jest zapis, czyli dane z magistrali 

writedata powinny zostać zapisane w miejsce wskazywane przez address.
writedata 32 In Magistrala zapisu danych

Pozostałe wymagane linie
clk 1 In Sygnał zegarowy. Przy zboczu opadającym inne linie powinny być stabilne – następuje wtedy zatrza-

skiwanie danych.
reset_n 1 In Wystąpienie stanu niskiego na tej linii powinno powodować reset modułu.

Tabela 2. Wyjaśnienie zachowania linii byteenable
Instrukcja zapisu byteenable address writedata Rejestr[0] Rejestr[1]

reset układu - - - 0x00000000 0x00000000
IOWR_32DIRECT(0, 0, 0x12345678); 0b1111 0b0 0x12345678 0x12345678 0x00000000
IOWR_32DIRECT(0, 4, 0x90ABCDEF); 0b1111 0b1 0x90ABCDEF 0x12345678 0x90ABCDEF
reset układu - - - 0x00000000 0x00000000
IOWR_16DIRECT(0, 0, 0x1234); 0b0011 0b0 0xXXXX1234 0x00001234 0x00000000
IOWR_16DIRECT(0, 2, 0x5678); 0b1100 0b0 0x5678XXXX 0x56781234 0x00000000
IOWR_16DIRECT(0, 4, 0x90AB); 0b0011 0b1 0xXXXX90AB 0x56781234 0x000090AB
IOWR_16DIRECT(0, 6, 0xCDEF); 0b1100 0b1 0xCDEFXXXX 0x56781234 0xCDEF90AB
reset układu - - - 0x00000000 0x00000000
IOWR_8DIRECT(0, 0, 0x12); 0b0001 0b0 0xXXXXXX12 0x00000012 0x00000000
IOWR_8DIRECT(0, 1, 0x34); 0b0010 0b0 0xXXXX34XX 0x00003412 0x00000000
IOWR_8DIRECT(0, 2, 0x56); 0b0100 0b0 0xXX56XXXX 0x00563412 0x00000000
IOWR_8DIRECT(0, 3, 0x78); 0b1000 0b0 0x78XXXXXX 0x78563412 0x00000000
IOWR_8DIRECT(0, 4, 0x90); 0b0001 0b1 0xXXXXXX90 0x78563412 0x00000090
IOWR_8DIRECT(0, 5, 0xAB); 0b0010 0b1 0xXXXXABXX 0x78563412 0x0000AB90
IOWR_8DIRECT(0, 6, 0xCD); 0b0100 0b1 0xXXCDXXXX 0x78563412 0x00CDAB90
IOWR_8DIRECT(0, 7, 0xEF); 0b1000 0b1 0xEFXXXXXX 0x78563412 0xEFCDAB90

z poziomu systemu NIOS. Ponadto, ważnym aspektem jest to, że dane 
muszą być odpowiednio wyrównane. Tzn. daną 32 bitową możemy 
zapisać pod adresem 0, 4, 8, 12, …, ale już nie pod adresami, które 
powodowałyby podzielenie 32-bitowego słowa na 2 adresy układu 
slave, czyli np. zapisy danej 32-bit pod adres 1, 3, 7, czy wszelkie inne 
nie będące podzielnymi przez 4 (czyli ilość bajtów w tym słowie) 
są niedopuszczalne. Zakładamy, że nasz moduł ma adres bazowy 
równy 0 (o tym jeszcze porozmawiamy za moment), a szerokość jego 
linii adresowej wynosi 1 bit (do dyspozycji zatem 2 słowa/rejestry 
32-bitowe). Instrukcje zapisu mają postać:

IOWR_nDIRECT(BASE, OFFSET, DATA),
gdzie:
•	 n to liczba bitów w zapisywanym słowie (n = 8, 16 lub 32),
•	 BASE to adres bazowy modułu,
•	 OFFSET to adres „rejestru” w danym module w adresowaniu 

z rozdzielczością bajtową,
•	 DATA to dana, którą chcemy zapisać.

W tabeli 2 umieszczono możliwe dla wspomnianej sytuacji użycia 
funkcji zapisu danych, wartości które przyjmie magistrala address 

sygnałem zegarowym – dane i adresy są zmieniane na zboczu 
narastającym, zaś zatrzaskiwane (stabilne) powinny być na zbo-
czu opadającym.

Myślę, że na tym etapie największa niewiadomą stanowi tajemni-
cza linia byteenable. Aby zrozumieć jej działanie zwróćmy uwagę 
na fakt, że po stronie układu master adresowanie odbywa się baj-
tami, zaś szerokość naszej magistrali to 32 bity, czyli 4 bajty. Dalej, 
biorąc pod uwagę to, co powtórzyłem za stosowną dokumentacją, 
że linia adresowa na wejściu układu slave pozwala na adresowanie 
słowami (czyli paczkami po 4 bajty). Z jednej strony dzięki takiemu 
rozwiązaniu możemy w pełni wykorzystać 32-bitową architekturę 
i szybciej przesyłać dane. Z drugiej strony powstaje jednak problem 
ze sposobem adresowania „mniejszych” porcji danych (czyli baj-
tów, czy danych 16-bitowych). Właśnie tu z odsieczą przybywają 
linie byteenable. Jak można zorientować się po pewnym czasie (lub 
przeczytać w dokumentacji), poszczególne bity odpowiadają za ak-
tywację zapisu odpowiadających im bajtów w 32-bitowym słowie.

Aby lepiej zrozumieć „jak to działa” weźmy pod uwagę moż-
liwe instrukcje zapisu na magistralę Avalon-MM, które są dostępne 

danego modułu, oraz magistrale writedata i, co najważniejsze, by-
teenable. Potem także zaprezentowane są  oczekiwane wartości 
32-bitowych rejestrów po wykonaniu każdej z operacji. X oznacza, 
że dane czwórki bitów nie powinny być brane pod uwagę w czasie 
danej operacji.

Mam nadzieję, że  spokojna analiza powyższych przykładów 
w pełni wyjaśnia działanie mechanizmu umożliwiającego adreso-
wanie zmiennych mniejszych niż 32 bity. 

W zasadzie pozostaje jeszcze teraz odpowiedź dotycząca adresowa-
nia całego modułu, czyli adresu bazowego. W naszym systemie adres 
ma postać 32-bitów, adresujących przestrzeń adresową bajt po baj-
cie. Stosując zatem 32-bitową szynę danych do adresowania takich 
słów potrzeba 30 bitów (zaś 2 najniższe bity mają odzwierciedlenie 
w byteenable). I w tym momencie te 30 bitów musi odpowiadać za 
adresowanie rejestrów w danym module, oraz za wybranie tylko jed-
nego modułu. Do wyboru modułu zatem brane musza być pod uwagę 
te bity, które nie adresują jego rejestrów – czyli dla wspomnianego 
wcześniej modułu, na którym „ćwiczyliśmy” zachowanie linii by-
teenable, na adres rejestru zostaną wykorzystane 4 bity, zatem aby 
zadecydować, czy dany moduł ma być aktywny czy nie, wykorzy-
stane zostanie 28 pozostałych najstarszych bitów adresu. I to war-
tość tych bitów definiuje adres bazowy modułu! Tylko i wyłącznie 
gdy będą one zgodne – dany moduł zostanie wybrany do prowadze-
nia komunikacji – w przeciwnym razie jego linia write nie będzie 
sterowana, a stan linii readdata będzie ignorowany.

Spójrzmy na powyższy przykład definiowanych przestrzeni ad-
resowych dla różnych modułów (rysunek 1). Dla pierwszej pozy-
cji, pamięci RAM, widzimy, że adresem bazowym jest 0x8000 oraz 
że ostatnim adresem jest 0xFFFF. Widzimy zatem, że dostępne mamy 
32768 adresów (bez zaskoczenia – tyle zadeklarowaliśmy jako roz-
miar pamięci RAM wyrażony w bajtach) – możemy to „zaadresować” 
za pomocą 14 bitów (od adresu 0 do adresu 32767 mamy 32768 „po-
zycji”). Zatem 18 najstarszych bitów będzie brane pod uwagę przy 
„decyzji” czy prowadzić komunikację z tą pamięcią ram czy nie. Jeśli 
bity tę będą zgodne z wartościami dla adresu bazowego – komuni-
kacja będzie „nawiązana”. W przeciwnym wypadku moduł pamięci 
RAM nie będzie „wiedział”, że na magistrali coś się dzieje. Na szczę-
ście o tę kwestię dba już sam Platfrom Designer, także poprzez opcję 
Assign Base Addresses, a poza tym lepiej będzie jak nauczymy się 
o tej kwestii w praktyce!

Implementacja modułu PWM
Przygotowania Na początek otwórzmy projekt, na którym ćwiczyli-
śmy działanie timera i wykorzystywaliśmy je do migania diodami. 
W Platform Designer usuńmy moduł PIO używany do sterowania 
diod LED. Dodatkowo w Pin Planner usuńmy wszystkie powiązane 
z tym modułem przypisane piny. Następnie, zanim zabierzemy się 
do projektowania naszego modułu PWM, możemy posłużyć się istnie-
jącym modułem pamięci RAM, aby zobaczyć graficzną reprezentację 
części opisu działania magi-
strali Avalon-MM. Aby to zro-
bić wystarczy, że klikniemy 
w Platform Designer na linijkę 
z portem s1 w bloku pamięci 
RAM (rysunek 2). Ukaże się 

na dole w okienku (w którym normalnie podglądamy właściwości 
komponentów) widok przebiegów czasowych sygnałów na magistrali 
Avalon-MM (rysunek 3). 

Występuje tu niezalecany do nowo tworzonych komponentów 
sygnał chipselect – u nas stosujemy sygnał read, który moglibyśmy 
„zapisać” jako: read = chipselect & !wirte. Poza tą drobną różnicą 
wynikająca z faktu, iż obecnie producent nie zaleca stosowania sy-
gnału chipselect, w naszym rozwiązaniu zastosujemy dokładnie taki 
schemat realizacji zapisu i odczytu. A to są jakieś inne? – zapewne 
zapytacie. Oczywiście! O czym przekonamy się w odpowiednim 
momencie, teraz nadmienię jedynie o tym, że możemy konfiguro-
wać m.in. dodatkowe opóźnienia pomiędzy istotnymi fragmentami 
sekwencji zapisu bądź odczytu, w sytuacji jeśli z jakichś powodów 
nasz układ nie nadążyłby z tak szybkim działaniem. Wróćmy jednak 
do tematu – co „dzieje się” na przedstawionych wykresach?

W wypadku zapisu, CPU ustawia adres, linie wyboru bajtów oraz 
dane do zapisu w czasie zbocza narastającego – nasz układ (tu pa-
mięć RAM) w czasie zbocza opadającego, gdy dane są stabilne, po-
winna je „zatrzasnąć” i przeanalizować, czyli w najpopularniejszej 
sytuacji – po prostu zapisać.

W sekwencji odczytu na pierwszym zboczu narastającym CPU 
ustawia adres do odczytu, a nasz moduł powinien ten adres zatrza-
snąć i przygotować daną, tak aby na najbliższym zboczu narasta-
jącym daną tą wystawić na odpowiednią szynę, aby CPU mógł ją 
odczytać przy drugim z kolei zboczu opadającym. Czyż to nie dzie-
cinnie łatwe?

Jak działa PWM? PWM, czyli modulacja szerokości impulsu (ang. 
Pulse Width Modulation), polega na tym, że sygnał cyfrowy o sta-
łym okresie zmienia swój współczynnik wypełnienia, czyli stosu-
nek czasu przez jaki trwa poziom wysoki do czasu całego okresu. 
W praktyce najprościej PWM zaimplementować jako licznik cyfrowy 
z dołączonym do niego komparatorem – jeśli wartość zadana jest 
większa (lub równa) niż wartość licznika – na wyjściu ustawiamy 
‘1’, w przeciwnym wypadku ‘0’. Przykładowe przebiegi PWM dla róż-
nego wypełnienia, wraz z licznikiem 0…9 (czyli fachowo mówiąc 
licznikiem modulo 10), pokazano na rysunku 4.

Oczywiście, jest to tylko jedna z możliwych implementacji PWM. 
Po pierwsze różnie można traktować porównanie (silne lub słabe – dla-
tego też zapisałem wyżej „lub równa” w nawiasie). Co więcej możemy 
w różny sposób implementować licznik – szczególnie ciekawym przy-
padkiem jest implementacja licznika jako liczącego w górę i w dół (tj. 
0-9, potem 9-0 itd.). Dzięki temu uzyskujemy tzw. phase correct PWM, 

Rysunek 1. Mapa adresów systemu

Rysunek 3. Przebiegi na magistrali Avalon-MM dla pamięci RAM

Rysunek 4. Przykładowe przebiegi PWM dla wypełnień 0, 10, 20 
i 90% (czyli odpowiadają temu wartości „zadane” – progi kompara-
cji – równe odpowiednio 0, 1, 2, 9

Rysunek 2. Wybranie linii zawiera-
jącej port Avalon-MM pamięci RAM

potrzebne, bądź usunąć te nieprzydatne, tym samym oszczędzając 
zasoby układu FPGA (listing 4).

W dalszej kolejności musimy obsłużyć zapis rejestrów poprzez magi-
stralę Avalon-MM. Rejestry o adresach 0..3 będą u mnie przechowywały 
16-bitową wartość współczynnika wypełnienia dla każdego z kanałów. 

czyli taki PWM, w którym wszystkie kanały (o ile jest ich więcej niż 
1) są zgodne w fazie i nie następuje efekt widoczny na powyższym ry-
sunku (w chwilach czasu 1, 11 oraz 21), polegający na włączaniu się 
wszystkich kanałów z wypełnieniem >0% w tym samym momencie.

Ruszamy do pracy – implementacja Implementację modułu PWM 
zacznijmy od dodania do projektu w Quartusie nowego pliku *.vhd. Ja 
zapisałem ten plik w nowym podfolderze PWM4 i nazwałem PWM4.
vhd. W nowo utworzonym pliku, po zadeklarowaniu użycia wymaga-
nych bibliotek, deklarujemy porty, jakie będzie miał nasz moduł, jak 
na listingu 1. Widzimy na nim deklarację wszystkich wspomnianych 
wcześniej sygnałów magistrali Avalon-MM, oraz 4 kanałów wyjścio-
wych PWM. Opis modułu rozpoczynamy od deklaracji wewnętrz-
nych rejestrów. Przypomnę jednak w tym miejscu dobitnie ważny 
fakt – VHDL to nie klasyczny język programowania – nie możemy 
robić tutaj „kalek” z tego czego nauczyliśmy się programując w ja-
kimkolwiek języku (np. C, czy C++) – VHDL to język opisu sprzętu, 
zatem za pomocą jego składni opisujemy de facto 
układ logiczny – połączenie elementów takich jak 
bramki, przerzutniki itp.

Wewnątrz  „architektury” (miejsce oznaczone 
(2)) definiujemy właściwa logikę działania. Za-
czynamy od zdefiniowania asynchronicznego (tj. 
nie zsynchronizowanego z zegarem) resetu, a do-
kładniej zachowania układu w momencie takiego 
resetu, polegającego na wyzerowaniu wszystkich 
ustawień i liczników. Definiujemy także zacho-
wanie układu w przypadku opadającego zbocza 
sygnału zegarowego (którą omówimy później) 
a także w przypadku narastającego zbocza zegara. 
Na  tym zboczu musimy 
jedynie zmieniać war-
tość magistrali readdata, 
zgodnie z  zależnościami 
czasowymi pokazanymi 
wcześniej na przykładzie 
pamięci RAM (patrz ry-
sunek 3). Z powodu faktu, 
że  adres zatrzaskiwany 
jest jednak na zboczu opa-
dającym musimy zastoso-
wać readdataBuffer, jak 
na listingu 3.

W  przypadku zbocza 
opadającego mamy dużo więcej 
do zrobienia. Po pierwsze musimy 
wygenerować sygnał PWM. Ponie-
waż chcemy aby nasz moduł był uni-
wersalny, wykonamy jeszcze dwie 
dodatkowe czynności. Po pierwsze, 
umożliwiamy definiowanie preska-
lera, czyli podzielnika powodującego, 
że nie przy każdym zboczu sygnału 
zegarowego zostanie zwiększona 
wartość licznika o którym mówili-
śmy przy okazji teoretycznego oma-
wiania działania PWM (rysunek 4). 
Po drugie, wartość maksymalną licz-
nika będziemy definiowali także za 
pomocą rejestru, dzięki czemu moż-
liwe będzie ustawienie takiej rozdzielczości, jaka akurat będzie nam 
potrzebna. Rzecz jasna powoduje to większe zużycie zasobów niż 
w wypadku zadeklarowania tych wartości w opisie VHDL na stałe 
– i tu, podobnie jak wspominałem przy okazji różnych opcji kon-
figuracji komponentów, możemy dopisać takie opcje jakie są nam 

Listing 1. Deklaracje wyprowadzeń
library ieee;
 
use ieee.std_logic_1164.all;
use ieee.std_logic_unsigned.all;
use ieee.numeric_std.all;
 
entity PWM4 is
    port ( 
                --avalon memory-mapped slave
                clk         : in std_logic;
                reset_n     : in std_logic;
                address     : in std_logic_vector(2 downto 0);
                byteenable  : in std_logic_vector(3 downto 0);
                read        : in std_logic;
                readdata    : out std_logic_vector(31 downto 0);
                write       : in std_logic;
                writedata   : in std_logic_vector(31 downto 0);
               
                --pwm exported interface
                pwm         : out std_logic_vector(3 downto 0)
 
        );
end PWM4;
...(1)...

Listing 2. Deklaracje bramek, przerzutników itp.
...(1)...
architecture basic of PWM4 is
    -- rejestr do zatrzaskiwania danych wyjściowych celem prawidłowej synchronizacji
    signal readdataBuffer   : std_logic_vector(31 downto 0);
    -- deklaracja "rejestru" przechowującego wartość preskalera
    signal counterPrescMax  : std_logic_vector(31 downto 0);
    -- deklaracja rejestru licznika preskalera
    signal counterPresc     : std_logic_vector(31 downto 0);
    -- deklaracja licznika do generowania PWM
    signal counterPWM           : std_logic_vector(15 downto 0);
    -- deklaracja rejestru maksymalnej wartości pwm (rozdzielczości)
    signal counterPWMMax        : std_logic_vector(15 downto 0);
    -- deklaracja typu tablicy i tablicy z elementami
    type PWMArrayType is array (0 to 3) of std_logic_vector(15 downto 0);
    signal PWMValue         : PWMArrayType;
begin
...(2)...
end basic;

Listing 3. Deklaracja bufora odczytywanych danych
...(2)...
process(clk, reset_n) is
    begin
        -- jeśli linia reset w stanie niskim
        if reset_n = '0' then
            counterPrescMax     <= (others => '0');
            counterPresc        <= (others => '0');
            counterPWM          <= (others => '0');
            counterPWMMax       <= (others => '0');
            PWMValue                <= (others=> (others=>'0'));
        -- jeśli opadające zbocze zegara
        elsif falling_edge(clk) then
            ...(3)...
        -- jeśli narastające zbocze zegara - zatrzaskujemy dane wyjściowe (zgodnie z oczekiwanym timingiem)
        elsif rising_edge(clk) then
            readdata <= readdataBuffer;
        end if;
    end process;

Listing 4. Utworzenie generatora PWM oraz komparatora
...(3)...
-- generowanie sygnałóW PWM
            -- obsługujemy podzielnik sygnału zegarowego / preskaler
            if counterPresc < counterPrescMax then
                counterPresc <= counterPresc + 1;
            else
                counterPresc <= (others => '0');
                -- obsługujemy licznik PWM z zachowaniem maksymlanej wartości
                if counterPWM < counterPWMMax then
                    counterPWM <= counterPWM + 1;
                else
                    counterPWM      <= (others => '0');
                end if;
                -- dla każdego z kanałów tworzymy "komparator" do generowania sygnału wyjściowego
                for i in 0 to 3 loop
                    if PWMValue(i) >= counterPWM then
                        pwm(i) <= '1';
                    else
                        pwm(i) <= '0';
                    end if;
                end loop;
            end if;
...(4)...

Rejestr o adresie 4 będzie 
przechowywać 32-bitową 
wartość podzielnika sy-
gnału zegarowego, zaś 
ostatni zapamięta maksy-
malną wartość licznika 
i zdefiniuje rozdzielczość. 
Zauważmy, że  w  poniż-
szym kodzie zaimple-
mentowałem obsługę linii 
byteenable, dzięki czemu 
możemy wykorzystać do-
wolne funkcje zapisu. 
Zapytać możecie jednak 
–  po  co  to  robić, skoro 
i tak zapisywać będziemy 
wartości 16-bitowe, lub 
zawsze możemy użyć in-
strukcji zapisu warto-
ści 32-bit i  nie martwić 
się o te linie. Oczywiście 
– macie rację! W dodatku 
takie rozwiązanie byłoby 
bardziej eleganckie, czy-
telne, zajmowało mniej 
zasobów. I najprawdopo-
dobniej, gdybym walczył o zasoby właśnie tak 
bym postąpił. Do czegoś wręcz przeciwnego zmo-
tywowała mnie jednak chęć wykorzystania de-
buggera i trybu podglądu oraz modyfikowania 
pamięci (czyli de facto zapisywania i odczyty-
wania danych z magistrali Avalon-MM w naszym 
systemie). Narzędzie to właśnie zawsze dokonuje 
zapisu bajt po bajcie i w przypadku braku im-
plementacji tej możliwości korzystanie z niego 
jest niemożliwe (oczywiście nikt nam tego nie 
zabroni, ale poziom wewnętrznej frustracji w mo-
mencie gdy z naszego układu odczytamy wartość 
zupełnie inną niż chcieliśmy zapisać i zaczniemy 
szukać błędu w naszym kodzie powinien nas sku-
tecznie zniechęcić do takich praktyk) – listing 5.

Na  sam koniec obsłużymy odczyt, 
tu na szczęście nie jest wykorzystywana szyna 
byteenable i jedyna pułapka to opisane wcze-
śniej buforowanie aby zapewnić zmianę da-
nych tylko na zboczu narastającym zegara, jak 
pokazano na listingu 6.

Na sam koniec wspomnę jeszcze o zastosowa-
nej strukturze for ... loop. Nie powoduje ona wy-
generowana pętli, która powtórzy się za każdym 
cyklem zegara określoną liczbę razy – powoduje 
ona, że zawarte w niej elementy zostaną powie-
lone, podobnie jakbyśmy do takiego celu zasto-
sowali jakieś makra preprocesora w języku C. 

Ubieramy nasz opis w VHDL w „bloczek” 
Platform Designer Po zapisaniu naszego pliku 
z opisem modułu musimy połączyć go z resztą 
systemu w Platform Designer. Aby to zrobić 
w panelu IP Catalog klikamy na New Compo-
nent… Naszym oczom ukazuje się okno Com-
ponent Editor i to w nim dokonamy wszelkich 
koniecznych czynności. 

Na początku, tak jak w wypadku drivera 
I2C, podajemy nazwę modułu. Jako Display Rysunek 5. Panel konfiguracji interfejsu Avalon-MM

Listing 5. Obsługa magistrali Avalon
...(4)...
-- obsługa magistrali Avalon-MM - zapis do układu z uwzględnieniem linii byteenable
            if write = '1' then
                -- zapis dla każdego z 4 rejestrów przechowujących wartość wypełnienia dla każdego z kanałów
                -- będą one miały adresy 0-3
                for i in 0 to 3 loop
                    if address = std_logic_vector(to_unsigned(i,address'length)) then
                        if byteenable(0) = '1' then
                            PWMValue(i)(7 downto 0) <= writedata(7 downto 0);
                        end if;
                        if byteenable(1) = '1' then
                            PWMValue(i)(15 downto 8) <= writedata(15 downto 8);
                        end if;
                    end if;
                end loop;
                -- pod adresem 4 preskaler używany celem zmiany częstotliwości PWM
                if address = std_logic_vector(to_unsigned(4,address'length)) then
                    if byteenable(0) = '1' then
                        counterPrescMax(7 downto 0) <= writedata(7 downto 0);
                    end if;
                    if byteenable(1) = '1' then
                        counterPrescMax(15 downto 8) <= writedata(15 downto 8);
                    end if;
                    if byteenable(2) = '1' then
                        counterPrescMax(23 downto 16) <= writedata(23 downto 16);
                    end if;
                    if byteenable(3) = '1' then
                        counterPrescMax(31 downto 24) <= writedata(31 downto 24);
                    end if;
                end if;
                -- pod adresem 5 wartość definiująca maksymalną wartość licznika, a zatem rozdzielczość PWM
                if address = std_logic_vector(to_unsigned(5,address'length)) then
                    if byteenable(0) = '1' then
                        counterPWMMax(7 downto 0) <= writedata(7 downto 0);
                    end if;
                    if byteenable(1) = '1' then
                        counterPWMMax(15 downto 8) <= writedata(15 downto 8);
                    end if;
                end if;
            end if;
...(5)...

Listing 6. Zmiana danych na narastającym zboczu zegara
...(5)...

-- obsługa magistrali Avalon-MM - odczyt z układu
            if read = '1' then
                for i in 0 to 3 loop
                    if address = std_logic_vector(to_unsigned(i,address'length)) then
                        readdataBuffer(15 downto 0) <= PWMValue(i);
                        readdataBuffer(31 downto 16) <= (others => '0');
                    end if;
                end loop;
                if address = std_logic_vector(to_unsigned(4,address'length)) then
                    readdataBuffer <= counterPrescMax;
                end if;
                if address = std_logic_vector(to_unsigned(5,address'length)) then
                    readdataBuffer(15 downto 0) <= counterPWMMax;
                    readdataBuffer(31 downto 16) <= (others => '0');
                end if;
                if address > std_logic_vector(to_unsigned(5,address'length)) then
                    readdataBuffer <= (others => '0');
                end if;
            end if;

Powyższy kod powoduje ustawienie preskalera, maksymalnej war-
tości licznika PWM oraz danych poszczególnych kanałów PWM. 
Zwróćmy uwagę na to, że korzystamy z instrukcji zapisu 32 bito-
wego, oraz że offset wyrażony jest w bajtach. Po uruchomieniu pro-
gramu diody na płytce powinny przybrać różne jasności – od zgaszonej 
do świecącej bardzo jasno. 

Rysunek 6. Ustawienia magistrali Avalon-MM dla projektu genera-
tora PWM, wraz z widocznymi oczekiwanymi przebiegami

Na pewno zwrócicie uwagę na fakt, że pomimo liniowo rosną-
cych wartości PWM jasność diod nie jest liniowa. Wiąże się 
to ściśle z tym, jak jasność jest postrzegana przez ludzkie oko 
– nieliniowo. Aby uzyskać efekt liniowej zmiany jasności należy 
stosować odpowiednio wyliczone wartości PWM. W układzie 
FPGA moglibyśmy się pokusić o implementację odpowiedniej 
tablicy w sprzęcie i stworzenie liniowego modułu PWM – będzie 
to zadaniem dla osób lepiej znających języki opisu sprzętu.

name możemy podać nieco dłuższą nazwę, jaka będzie wyświetlana 
w różnych miejscach naszego systemu. Potem dodajemy nasz plik 
PWM4.vhd w zakładce Files do wszystkich okienek, po czym kli-
kamy na Analyze Synthesis Files i przechodzimy do zakładki Signals 
& Interfaces. Widzimy, że system nie najgorzej poradził sobie z przy-
pisaniem poszczególnych sygnałów (gdyż użyliśmy nazw zgodnych 
z jego konwencją). Problem wystąpił jednak z portem pwm (który zo-
stał włączony u mnie do magistrali Avalon) oraz z brakiem przypisa-
nia sygnału resetu do magistrali Avalon. 

Pierwszy z tych błędów (który spowodował ostrzeżenie i kilka in-
nych błędów) naprawiamy dodając interfejs (<<add interface>>) typu 
conduit i do niego przeciągając port pwm. Dokładnie tak samo jak przy 
tworzeniu podobnego interfejsu dla kontrolera I2C. Jako typ sygnału 
(Signal Type) możemy podać dowolną wartość, np. pwm czy conduit. 

Aby skonfigurować interfejs Avalon i usunąć ostatni z błędów kli-
kamy na avalon_slave_0 (nazwę tę możemy zmienić na bardziej przy-
jazną). Mamy teraz do dyspozycji panel konfiguracji parametrów 
interfejsu Avalon-MM (rysunek 5).

Na początku mamy do wyboru, oprócz nazwy interfejsu, która ma 
znaczenie jedynie czysto estetyczne, typ tego interfejsu (którego nie 
zmieniamy), powiązane z interfejsem sygnały zegarowy i resetu oraz 
dodatkowe opcje. Ponieważ ostatni błąd zwracał nam uwagę na brak 
przypisania sygnału resetu do interfejsu, właśnie w tym miejscu wy-
bieramy jedyny dostępny sygnał reset z listy rozwijanej. Gdyby poja-
wiły się jakiekolwiek inne problemy związane z interfejsem zawsze 
wybierajmy Signal Type zgodny z nazwą sygnału.

W kolejnym bloku ustawień (Parameters) do wyboru mamy opcje 
dotyczące sposobu adresowania oraz zdublowane definicje sygnału 
zegarowego i resetu – te pozostawiamy bez zmian, gdyż muszą być 
kompatybilne z tymi, jakie wykorzystuje NIOS II. 

Bloki ustawień Timing oraz Pipelined Transfers są najważniejszymi 
blokami – w nich musimy zdefiniować takie parametry, które odpo-
wiadają sposobowi obsługi magistrali, który zaimplementowaliśmy. 
Na szczęście nie musimy wszystkich tych opcji poznawać teoretycznie 
– po rozwinięciu najniższych bloków będziemy mieć dostęp do ściągi 
– oprogramowanie pokaże nam jak powinny wyglądać przebiegi dla 
każdego z wprowadzonych ustawień. Zachęcam zatem do poekspe-
rymentowania z tymi ustawieniami (najlepiej w tylko jednym polu 
wpisać wartość różną od 0 na początek), aby zrozumieć za co odpowie-
dzialny jest każdy z parametrów. Następnie, aby zapewnić ustawie-
nia kompatybilne z układem logicznym, jaki zaimplementowaliśmy, 
zerujemy wszystkie parametry poza Read latency, zaś ten parametr 
ustawiamy, jak pokazano na rysunku 6.

Po wprowadzeniu tych ustawień klikamy na Finish…, potwier-
dzamy chęć zapisania komponentu i… voila! Nasz blok PWM poja-
wia się na liście dostępnych komponentów! Teraz możemy go dodać 
do systemu, wykonać konieczne połączenia, wyeksportować sygnał 
pwm i wykonać analizę projektu, a następnie przypisać piny pwm 
do diod LED na płytce i wreszcie wykonać kompilację całego projektu.

Programujemy obsługę własnego modułu
Po pełnej kompilacji projektu i wgraniu go do układu FPGA możemy 
przystąpić do programowania. Jak już się zapewne domyślacie skorzy-
stamy ze wspomnianych wcześniej funkcji do zapisu (i ewentualnego 
odczytu) wartości z poszczególnych rejestrów modułu:
IOWR_32DIRECT(PWM_BASE, 16, 1); // preskaller 1 -> 
podział sygnału zegarowego na 2
IOWR_32DIRECT(PWM_BASE, 20, 4095); // maksymalna 
wartośc PWm - rozdzielczośc
//wartości poszczególnych kanałów
IOWR_32DIRECT(PWM_BASE, 0, 0);
IOWR_32DIRECT(PWM_BASE, 4, 2048);
IOWR_32DIRECT(PWM_BASE, 8, 3072);
IOWR_32DIRECT(PWM_BASE, 12, 4095); 

Chciałbym jednak teraz zaprezentować Wam jeszcze jedno ciekawe 
narzędzie do debuggowania. W pliku system.h (lub w Platform Desi-
gner) sprawdźmy jaki jest adres bazowy modułu PWM, a następnie 
uruchommy debugowanie projektu. Uruchommy program, a następnie 
kliknijmy na przycisk wstrzymania (pauzy) programu, po czym na ekra-
nie znajdźmy okienko Memory. Jeśli nie jest ono widoczne musimy je wy-
świetlić za pomocą menu Window  Show view  Memory. Teraz obok 
Monitor kliknijmy na zielony plus. W okienku, które pojawi się wpisu-
jemy adres bazowy modułu PWM (u mnie 0x11000) i zatwierdzamy ten 
wybór. W oknie powinien pojawić się widok pokazany na rysunku 7.

Jest to okno podglądu pamięci, ale pamiętając, że nasz moduł ko-
rzysta z magistrali mapującej wszelkie rejestry jako pewne adresy 
w pamięci, możemy powiedzieć, że to okienko pozwala nam na pod-
gląd m.in. rejestrów naszego modułu. Jeśli dobrze przeanalizujemy 
pierwsze wiersze tabeli to odnajdziemy tam nieco pomieszane, ale 
prawidłowe wartości, które ustawił nasz prosty program. W celu 
bardziej przejrzystej prezentacji danych możemy wybrać New Ren-
derings… a tam wybrać opcję Unsigned Integer (rysunek 8). Wtedy 
zawartość pamięci zostanie zinterpretowana jako normalne liczby. 

Co najciekawsze, narzędzie to pozwala nie tylko na odczyt danych, 
ale także na ich modyfikowanie! Spróbujcie wpisać np. pod adresem 
0x11000 wartość 4095, a po naciśnięciu ENTER jedna z diod powinna 
zgasnąć. Dzięki temu możemy bez pisania programu testować nasze 
moduły, o ile oczywiście zaimplementowaliśmy linie byteenable, 

Rysunek 8. Inna reprezentacja zawartości pamięci

Rysunek 9. Edycja definiowalnych przez użytkownika parametrów 
modułu

Listing 7. Wykonanie 4-kanałowego generatora PWM
entity PWMn is
    generic (
                CHANNELS        : integer := 4;
                ADDRBITS        : integer := 3
    );
    port ( 
                --avalon memory-mapped slave
                ...
                address     : in std_logic_vector(ADDRBITS-1 downto 0);
                ...               
                --pwm exported interface
                pwm         : out std_logic_vector(CHANNELS-1 downto 0)
 
        );
end PWMn;

Rysunek 7. Okno podglądu pamięci

Rysunek 10. Prawidłowo skonfigurowane szerokości magistral

bez których ta opcja debuggera działa w błędny spo-
sób jeśli chodzi o zapis danych. Nie przejmujcie się też 
tym, że wartość największa oznacza zgaszenie diody 
– po pierwsze wynika to z takiego a nie innego zasil-
nia diod, a po drugie… możemy to poprawić sprzętowo, 
negując sygnały w odpowiednich miejscach projektu 
w  VHDL, lub stosując „odwrócone” wartości w  na-
szych programach.

A może konfiguracja do tego?
Jakkolwiek poczyniliśmy już wielki krok ku wykorzystaniu możliwo-
ści układu FPGA, nadal nasze rozwiązanie przykładowe ma poważną 
wadę – mianowicie jest wysoce specjalizowane – umożliwia dodanie 
tylko i wyłącznie 4-kanałowego generatora PWM. Jeśli chcielibyśmy ka-
nałów więcej lub mniej skończyłoby się to na konieczności tworzenia 
kolejnego modułu z wymagana ilością kanałów. Po co jednak utrudniać 
sobie życie, skoro można skorzystać z dodatkowych możliwości języka 
VHDL (lub innego) i stworzyć generator PWM o definiowanej ilości ka-
nałów. Zatem do dzieła!

Rozpocznijmy od stworzenia i zapisania nowego pliku (u mnie PWMn/
PWMn.vhd) i przekopiowania do niego zawartości z wcześniej opraco-
wanego modułu PWM4. Następnie musimy dokonać w kodzie drobnych 
zmian, które umożliwią definiowanie ilości wyjść w jednym miejscu. 
Dodajemy w tym celu specjalną deklarację do naszego komponentu, 
która pozwala na przekazanie wymaganych parametrów. Przekazywać 
będziemy zatem ilość wejść oraz (niestety) ręcznie wyliczoną ilość bitów 
adresu (czyli ilość bitów koniecznych do zaadresowania 
rejestru każdego kanału oraz dwóch rejestrów, definiują-
cych częstotliwość pracy i rozdzielczość) – listing 7.

Ponadto dokonujemy także zmian w kodzie, aby wy-
generować za pomocą pętli odpowiednią ilość kanałów, 
oraz aby nadać odpowiednie adresy rejestrom. Przykła-
dowe zmiany to:
for i in 0 to CHANNELS-1 loop
...
if address = std_logic_vector(to_
unsigned(CHANNELS+1,address'length)) then

Następnie wykonujemy procedurę dodawania nowego modułu, 
nadając mu rzecz jasna inną nazwę i także korygujemy identyczne 
błędy, które mogą powstać i tym razem. Otwórzmy teraz zakładkę 
Parameters (rysunek 9).

Możemy tutaj wyłączyć możliwość edycji niektórych parame-
trów (kolumna Edit…), pogrupować je oraz dodać wskazówki dla 
każdego z pól, co zrobiłem. Po kliknięciu na Preview the GUI... 
możemy zobaczyć jak będzie wyglądało dostępne później menu 
edycji tych parametrów.

W kolejnym kroku musimy sprawić, aby także Platform Desi-
gner wiedział, że szerokość niektórych portów będzie zmieniać się w za-
leżności od tych parametrów. W tym celu klikamy na View  Signals, 
dzięki czemu pokaże się nam nowa karta z zebraną listą wszelkich wy-
prowadzeń modułu (są to dane zdublowane z zakładki Signals & Inter-
faces, jednak w obecnej wersji oprogramowania występuje tam problem 
z wyświetlaniem szerokości magistral zależnych od parametrów – cały 
czas wtedy wyświetlana jest wartość „–1”). W kolumnie Width wpisu-
jemy dla magistrali address nazwę odpowiedniego parametru, a więc 
ADDRBITS, zaś dla pwm będzie to CHANNELS. Ostatecznie lista po-
winna wyglądać w sposób pokazany na rysunku 10.

Klikamy na Finish i zapisujemy nasz komponent. Następnie pozostało 
nam już tylko usunąć z projektu stary moduł PWM ze stałą ilością ka-
nałów i zastąpić go nową, ulepszoną wersją. Jeśli zaś chcielibyśmy, aby 
ilość bitów wyliczana była automatycznie na podstawie liczby kana-
łów, musielibyśmy sięgnąć do skryptu tcl, który jest generowany przez 
Platform Designer dla każdego nowo tworzonego modułu i tam dopisać 
instrukcje automatyzujące ten proces.

Podsumowanie
Nauczyliśmy się dziś bardzo ważnych umiejętności, przede wszyst-
kim poznaliśmy magistralę Avalon-MM i samodzielnie wykonali-
śmy wykorzystujące ją komponenty, możliwe do dodania za pomocą 
oprogramowania Platform Designer. W ramach zadania domowego 
polecam poeksperymentować z wykorzystaniem modułu PWM 
– spróbować podłączyć do zestawu więcej diod LED, dodać mo-
duł PWM z odpowiednią liczbą wyjść (oraz właściwą liczbą wejść 
adresowych) i napisać program, który sterowałby nimi w ciekawy 
sposób – wszak od migania diodą każdy z nas zaczynał kiedyś. Po-
nadto przypomnijcie sobie koniecznie zasadę multipleksowania 
wyświetlacza 7-segmentowego – w czasie kolejnego spotkania napi-
szemy moduł realizujący tę czynność sprzętowo – bez zaprzątania 
tym cennego czasu procesora. Zaopatrzcie się także w praktycz-
nie dowolny enkoder inkrementalny, gdyż i  jego obsługę zreali-
zujemy sprzętowo.

Piotr Rzeszut, AGH


